# (c) 2009 Nokia Corporation
QT += webkit \
    sql \
    network 

TARGET = cartoonreader
VERSION = 1.2.0

TEMPLATE = app
SOURCES += src/main.cpp \
    src/cartoonreader.cpp \
    src/windowwrapper.cpp \
    src/platformresolver.cpp
HEADERS += src/cartoonreader.h \
    src/windowwrapper.h \
    src/platformresolver.h
RESOURCES += cartoonreader.qrc
Debug:DEFINES += DEBUG

# Symbian-specific
symbian: { 
    TARGET.UID3 = 0xEdbca140
    TARGET.CAPABILITY = NetworkServices
    TARGET.EPOCHEAPSIZE = 0x100000 \
        0x2500000
    TARGET.EPOCSTACKSIZE = 0x14000
    ICON = images/symbian/$${TARGET}.svg
	
    # Libraries needed for locking Symbian device's orientation to landscape
    LIBS += -lcone \
            -leikcore \
            -lavkon
}

# Maemo specific
maemo5: { 
    # VARIABLES
    isEmpty(PREFIX):PREFIX = /usr
    BINDIR = $$PREFIX/bin
    DATADIR = $$PREFIX/share
    DEFINES += DATADIR=\\\"$$DATADIR\\\" \
        PKGDATADIR=\\\"$$PKGDATADIR\\\"
    
    # MAKE INSTALL
    INSTALLS += target \
        desktop \
        iconxpm \
        icon26 \
        icon40 \
        icon64
    target.path = $$BINDIR
    desktop.path = $$DATADIR/applications/hildon
    desktop.files += src/$${TARGET}.desktop
    iconxpm.path = $$DATADIR/pixmap
    iconxpm.files += images/maemo/$${TARGET}.xpm
    icon26.path = $$DATADIR/icons/hicolor/26x26/apps
    icon26.files += images/maemo/26x26/$${TARGET}.png
    icon40.path = $$DATADIR/icons/hicolor/40x40/apps
    icon40.files += images/maemo/40x40/$${TARGET}.png
    icon64.path = $$DATADIR/icons/hicolor/64x64/apps
    icon64.files += images/maemo/64x64/$${TARGET}.png
}
